The macro of mode-I edge crack under uniform tension loading for Circular zone model

Author: İ. Y. Sülü, A. O. Ayhan





!!! This macro is used to solve mode-I edge crack problems in a

!!! semi-infinite strip under uniform tension loadings.

!!! The parameters/variables taken by the macro are:
!!! P : Uniform tension loading
!!! CR_L : Crack length
!!! WW : Plate width
!!! HH: 1/2 height of plate
!!! NEL_THET : Elements numbers at θ angle direction
!!! NEL_R : Elements numbers at radial direction
!!! CONSTANT : Crack length to plate width ratio
!!! CRFACE_LESIZE : Elements distribution ratio in the crack surface
!!! R_SIZE_RATIO : Distance ratio between elements at radial direction of crack zone
!!! R_F : Size of first circular element in the crack zone
!!! R_F1 : Size of three circular elements in the crack zone

!!! R_C : Size of crack zone at radial direction

```
P=1

CR_L=1

CONSTANT=0.7

WW= CR_L /CONSTANT

HH=40

NEL_THET=9

NEL_R=8

CRFACE_LESIZE=0.2

R_SIZE_RATIO=2

R_F=0.03

R_F1=R_F*3

R_C=R_F1+(NEL_R-3)*R_F* R_SIZE_RATIO

TO_1= CR_L/10E6

TO=CR_L- TO_1

/PREP7

ET,1,PLANE82

MP,EX,1,30E4

MP,NUXY,1,0.3

K,1,

K,2,,HH

K,3,WW,HH

K,4,WW

CYL4,CR_L,,,,R_F1,180

CYL4,CR_L,, R_F1,,R_C,180

NUMMRG,KP

L,9,1

L,1,2
```

```
L,2,3

L,3,4

L,4,8

AL,4,6,8,9,10,11

APLOT

ASEL,NONE

LSEL,NONE

KSEL,NONE

K,10,

K,11,,-HH

K,12,WW,-HH

K,13,WW

CYL4,CR_L,,,180,R_F1,360

CYL4,CR_L,,R_F1,180,R_C,360

NUMMRG,KP

L,17,10

L,10,11

L,11,12

L,12,13

L,13,18

AL,15,17,19,20,21,22

APLOT

ALLSEL

KSEL,,LOC,Y,- TO_1, TO_1

KSEL,U,LOC,X,- TO_1, TO

NUMMRG,KP

ALLSEL,,KP
```

```
APLOT

ESIZE,0.3

KWPAVE,      7

WPRO,,,90.000000

FLST,2,4,5,ORDE,4

FITEM,2,1

FITEM,2,-2

FITEM,2,4

FITEM,2,-5

ASBW,P51X

LESIZE,6, CRFACE_LESIZE

LESIZE,17, CRFACE_LESIZE

KSCON,7, R_F,1, NEL_THET,1

LESIZE,3,,,3

LESIZE,24,,,3

LESIZE,16,,,3

LESIZE,2,,,3

LESIZE,14,,,3

LESIZE,7,,, (NEL_R-3)

LESIZE,30,,, (NEL_R-3)

LESIZE,27,,, (NEL_R-3)

LESIZE,5,,, (NEL_R-3)

LESIZE,18,,, (NEL_R-3)

LESIZE,23,,,NEL_THET

LESIZE,22,,,NEL_THET

LESIZE,13,,,NEL_THET

LESIZE,25,,,NEL_THET
```

```
LESIZE,26,,,NEL_THET
LESIZE,28,,,NEL_THET
LESIZE,29,,,NEL_THET
LESIZE,31,,,NEL_THET
AMESH,8
AMESH,7
AMESH,9
AMESH,10
AMESH,3
AMESH,6
MSHKEY,1
AMESH,12
MSHKEY,1
AMESH,11
MSHKEY,1
AMESH,13
MSHKEY,1
AMESH,14
DK,12,UY,0
DK,11,UX,0
DK,11,UY,0
SFL,9,PRES,-P
SFL,20,PRES,-P
SAVE
/SOLU
SOLVE
/POST1
```

```
PLNSOL,S,Y
/EDGE,1,1
REPLOT
CLOCAL,11,1,CR_L,0,0
NSEL,,LOC,X,- CR_L/10E6, CR_L/10E6
*GET,NN1,NODE,0,NUM,MAX
NSEL,,LOC,Y,180+0.01,180-180/NEL_THET-0.01
NSEL,R,LOC,X,-0.001,R_F+R_F/100
NSEL,A,LOC,X,-0.001,0.001
ESLN,,1
NSEL,R,LOC,X,R_F/4-R_F/100,R_F/4+R_F/100
NSEL,R,LOC,Y,180+0.01,180-0.01
NSLE,R
*GET,NN2,NODE,0,NUM,MAX
NSEL,,LOC,Y,180+0.01,180-180/NEL_THET-0.01
NSEL,R,LOC,X,-0.001,R_F+R_F/100
NSEL,A,LOC,X,-0.001,0.001
ESLN,,1
NSEL,R,LOC,X,R_F-R_F/100,R_F+R_F/100
NSEL,R,LOC,Y,180+0.01,180-0.01
NSLE,R
*GET,NN3,NODE,0,NUM,MAX
NSEL,,LOC,Y,-180-0.01,-180+180/NEL_THET+0.01
NSEL,R,LOC,X,-0.001,R_F+R_F/100
NSEL,A,LOC,X,-0.001,0.001
ESLN,,1
NSEL,R,LOC,X,R_F/4-R_F/100,R_F/4+R_F/100
```

```
NSEL,R,LOC,Y,-180-0.01,-180+0.01
NSLE,R
*GET,NN4,NODE,0,NUM,MAX
NSEL,,LOC,Y,-180-0.01,-180+180/NEL_THET+0.01
NSEL,R,LOC,X,-0.001,R_F+R_F/100
NSEL,A,LOC,X,-0.001,0.001
ESLN,,1
NSEL,R,LOC,X,R_F-R_F/100,R_F+R_F/100
NSEL,R,LOC,Y,-180-0.01,-180+0.01
NSLE,R
*GET,NN5,NODE,0,NUM,MAX
CSYS,0
/POST1
ALLSEL
PATH,K1,5,,30,20
PPATH,1,NN1
PPATH,2,NN2
PPATH,3,NN3
PPATH,4,NN4
PPATH,5,NN5
KCALC,1,1,3
```

!!! KCALC : Calculation of stress intensity factor (K1)